
Abstract. We consider a simplified version of Nakano’s guarded fixed-point
types in a representation by infinite type expressions, defined coinductively. Small-
step reduction is parametrized by a natural number “depth” that expresses under
how many guards we may step during evaluation. We prove that reduction is
strongly normalizing for any depth. The proof involves a typed inductive notion
of strong normalization and a Kripke model of types in two dimensions: depth
and typing context. Our results have been formalized in Agda and serve as a case
study of reasoning about a language with coinductive type expressions.

1 Introduction

In untyped lambda calculus, fixed-point combinators can be defined using self-appli-
cation. Such combinators can be assigned recursive types, albeit only negative ones.
Since such types introduce logical inconsistency, they are ruled out in Martin-Löf Type
Theory and other systems based on the Curry-Howard isomorphism. Nakano (2000)
introduced a modality for recursion that allows a stratification of negative recursive
types to recover consistency. In essence, each negative recursive occurrence needs to
be guarded by the modality; this coined the term guarded recursive types (Birkedal
and Møgelberg, 2013).1 Nakano’s modality has found applications in functional reac-
tive programming (Krishnaswami and Benton, 2011b) where it is referred to as later
modality.

While Nakano showed that every typed term has a weak head normal form, in this
paper we prove strong normalization for our variant λI of Nakano’s calculus. To this
end, we make the introduction rule for the later modality explicit in the terms by a
constructor next, following Birkedal and Møgelberg (2013) and Atkey and McBride
(2013). By allowing reduction under finitely many nexts, we establish termination ir-
respective of the reduction strategy. Showing strong normalization of λI is a first step
towards an operationally well-behaved type theory with guarded recursive types, for
which Birkedal and Møgelberg (2013) have given a categorical model.

Our proof is fully formalized in the proof assistant Agda (2014) which is based on
intensional Martin-Löf Type Theory. 2 One key idea of the formalization is to represent

1 Not to be confused with Guarded Recursive Datatype Constructors (Xi et al., 2003).
2 A similar proof could be formalized in other systems supporting mixed induction-coinduction,

for instance, in Coq.

the recursive types of λI as infinite type expressions in form of a coinductive defi-
nition. For this, we utilize Agda’s new copattern feature (Abel et al., 2013). The set
of strongly normalizing terms is defined inductively by distinguishing on the shape of
terms, following van Raamsdonk et al. (1999) and Joachimski and Matthes (2003). The
first author has formalized this technique before in Twelf (Abel, 2008); in this work we
extend these results by a proof of equivalence to the standard notion of strong normal-
ization.

Due to space constraints, we can only give a sketch of the formalization; a longer
version and the full Agda proofs are available online (Abel and Vezzosi, 2014). This
paper is extracted from a literate Agda file; all the colored code in displays is necessarily
type-correct.

2 Guarded Recursive Types and Their Semantics

Nakano’s type system (2000) is equipped with subtyping, but we stick to a simpler
variant without, a simply-typed version of Birkedal and Møgelberg (2013), which we
shall call λI. Our rather minimal grammar of types includes product A×B and function
types A→ B, delayed computations IA, variables X and explicit fixed-points µXA.

A,B,C ::= A×B | A→ B |IA | X | µXA

Base types and disjoint sum types could be added, but would only give breadth rather
than depth to our formalization. As usual, a dot after a bound variable shall denote
an opening parenthesis that closes as far to the right as syntactically possible. Thus,
µX .X → X denotes µX (X → X), while µXX → X denotes (µX .X)→ X (with a free
variable X).

Formation of fixed-points µXA is subject to the side condition that X is guarded
in A, i. e., X appears in A only under a later modality I. This rules out all unguarded
recursive types like µX .A×X or µX .X → A, but allows their variants µX .I(A×X)
and µX .A×IX , and µX .I(X → A) and µX .IX → A. Further, fixed-points give rise
to an equality relation on types induced by µXA = A[µXA/X ].

Γ (x) = A
Γ ` x : A

Γ ,x:A ` t : B
Γ ` λx. t : A→ B

Γ ` t : A→ B Γ ` u : A
Γ ` t u : B

Γ ` t1 : A1 Γ ` t2 : A2

Γ ` (t1, t2) : A1×A2

Γ ` t : A1×A2

Γ ` fst t : A1

Γ ` t : A1×A2

Γ ` snd t : A2

Γ ` t : A
Γ ` next t :IA

Γ ` t :I(A→ B) Γ ` u :IA
Γ ` t ∗u :IB

Γ ` t : A A = B
Γ ` t : B

Fig. 1. Typing rules.

Terms are lambda-terms with pairing and projection plus operations that witness
applicative functoriality of the later modality (Atkey and McBride, 2013).

t,u ::= x | λxt | t u | (t1, t2) | fst t | snd t | next t | t ∗u

Figure 1 recapitulates the static semantics. The dynamic semantics is induced by the
following contractions:

(λx. t)u 7→ t[u/x]
fst (t1, t2) 7→ t1
snd (t1, t2) 7→ t2
(next t)∗(next u) 7→ next (t u)

If we conceive our small-step reduction relation −→ as the compatible closure of 7→,
we obtain a non-normalizing calculus, since terms like Ω = ω (next ω) with ω =
(λx. x∗(next x)) are typeable.3 Unrestricted reduction of Ω is non-terminating: Ω −→
next Ω −→ next (next Ω) −→ . . . If we let next act as delay operator that blocks re-
duction inside, we regain termination. In general, we preserve termination if we only
look under delay operators up to a certain depth. This can be made precise by a family
−→n of reduction relations indexed by a depth n ∈ N, see Figure 2.

t 7→ t ′

t −→n t ′
t −→n t ′

λx. t −→n λx. t ′
t −→n t ′

t u−→n t ′ u
u−→n u′

t u−→n t u′

t −→n t ′

(t,u)−→n (t ′,u)
u−→n u′

(t,u)−→n (t,u′)
t −→n t ′

fst t −→n fst t ′
t −→n t ′

snd t −→n snd t ′

t −→n t ′

next t −→n+1 next t ′
t −→n t ′

t ∗u−→n t ′ ∗u
u−→n u′

t ∗u−→n t ∗u′

Fig. 2. Reduction

We should note that for a fixed depth n the relation−→n is not confluent. In fact the
term (λ z.nextn+1 z)(fst (u, t)) reduces to two different normal forms, nextn+1 (fst (u, t))
and nextn+1 u. We could remedy this situation by making sure we never hide redexes
under too many applications of next and instead store them in an explicit substitution
where they would still be accessible to −→n. Our problematic terms would then look
like nextn ((next z)[fst (u, t)/z]) and nextn ((next z)[u/z]) and the former would reduce
to the latter. However, we are not bothered by the non-confluence since our semantics
at level n (see below) does not distinguish between nextn+1u and nextn+1u′ (as in u′ =
fst (u, t)); neither u nor u′ is required to terminate if buried under more than n nexts.

To show termination, we interpret types as sets A ,B,C of depth-n strongly nor-
malizing terms. We define semantic versions J×K, J→K, and JIK of product, function

3 `Ω : A with A = µX(IX). To type ω , we use x : µY (I(Y → A)).

space, and delay type constructor, plus a terminal (=largest) semantic type J>K. Then
the interpretation JAKn of closed type A at depth n can be given recursively as follows,
using the Kripke construction at function types:

JA×BKn = JAKn J×K JBKn A J×K B = {t | fst t ∈A and snd t ∈B}
JA→ BKn =

⋂
n′≤n(JAKn′ J→K JBKn′) A J→K B = {t | t u ∈B for all u ∈A }

JIAK0 = JIKJ>K J>K = {t | t term}
JIAKn+1 = JIKJAKn JIKA = {next t | t ∈A }
JµXAKn = JA[µXA/X ]Kn (A is weak head expansion closure of A )

Due to the last equation (µ), the type interpretation is ill-defined for unguarded recur-
sive types. However, for guarded types we only return to the fixed-point case after we
have passed the case for I , which decreases the index n. More precisely, JAKn is de-
fined by lexicographic induction on (n,size(A)), where size(A) is the number of type
constructor symbols (×,→, µ) that occur unguarded in A.

While all this sounds straightforward at an informal level, formalization of the de-
scribed type language is quite hairy. For one, we have to enforce the restriction to well-
formed (guarded) types. Secondly, our type system contains a conversion rule, getting
us into the vincinity of dependent types which are still a challenge to a completely for-
mal treatment (McBride, 2010). Our first formalization attempt used kinding rules for
types to keep track of guardedness for formation of fixed-point, and a type equality
relation, and building on this, inductively defined well-typed terms. However, the com-
plexity was discouraging and lead us to a much more economic representation of types,
which is described in the next section.

3 Formalized Syntax

In this section, we discuss the formalization of types, terms, and typing of λI in Agda.
It will be necessary to talk about meta-level types, i. e., Agda’s types, thus, we will refer
to λI’s type constructors as ×̂, →̂, Î , and µ̂ .

3.1 Types Represented Coinductively

Instead of representing fixed-points as syntactic construction on types, which would re-
quire a non-trivial equality on types induced by µ̂XA = A[µ̂XA/X ], we use meta-level
fixed-points, i. e., Agda’s recursion mechanism.4 Extensionally, we are implementing
infinite type expressions over the constructors ×̂, →̂, and Î . The guard condition on re-
cursive types then becomes an instance of Agda’s “guard condition”, i. e., the condition
the termination checker imposes on recursive programs.

4 An alternative to get around the type equality problem would be iso-recursive types, i. e.,
with term constructors for folding and unfolding of µ̂XA. However, we would still have to
implement type variables, binding of type variables, type substitution, lemmas about type sub-
stitution etc.

Viewed as infinite expressions, guarded types are regular trees with an infinite num-
ber of Î -nodes on each infinite path. This can be expressed as the mixed coinductive(ν)-
inductive(µ) (meta-level) type

νXµY. (Y ×Y )+(Y ×Y )+X .

The first summand stands for the binary constructor ×̂, the second for →̂, and the third
for the unary Î . The nesting of a least-fixed point (µ) inside a greatest fixed-point (ν)
ensures that on each path, we can only take alternatives ×̂ and →̂ a finite number of
times before we have to choose the third alternative Î and restart the process.

In Agda 2.4, we represent this mixed coinductive-inductive type by a datatype Ty
(inductive component) mutually defined with a record ∞Ty (coinductive component).

mutual
data Ty : Set where
_×̂_ : (a b : Ty) → Ty
_→̂_ : (a b : Ty) → Ty
Î_ : (a∞ : ∞Ty) → Ty

record ∞Ty : Set where
coinductive
constructor delay_
�eld force_ : Ty

While the arguments a and b of the infix constructors ×̂ and →̂ are again in Ty, the
prefix constructor Î expects and argument a∞ in ∞Ty, which is basically a wrapping5

of Ty. The functions delay and force convert back and forth between Ty and ∞Ty so that
both types are valid representations of the set of types of λI.

delay : Ty→ ∞Ty
force : ∞Ty→ Ty

However, since ∞Ty is declared coinductive, its inhabitants are not evaluated until
forced. This allows us to represent infinite type expressions, like top= µ̂X(ÎX).

top : ∞Ty
force top = Î top

Technically, top is defined by copattern matching (Abel et al., 2013); top is uniquely
defined by the value of its only field, force top, which is given as Î top. Agda will use
the given equation for its internal normalization procedure during type-checking. Alter-
natively, we could have tried to define top : Ty by top = Îdelay top. However, Agda
will rightfully complain here since rewriting with this equation would keep expanding
top forever, thus, be non-terminating. In contrast, rewriting with the original equation
is terminating since at each step, one application of force is removed.

The following two defined type constructors will prove useful in the definition of
well-typed terms to follow.

5 Similar to a newtype in the functional programming language Haskell.



6

I_ : Ty → Ty
I a = Î delay a

_⇒_ : (a∞ b∞ : ∞Ty) → ∞Ty
force (a∞ ⇒ b∞) = force a∞ →̂ force b∞

3.2 Well-typed terms

Instead of a raw syntax and a typing relation, we represent well-typed terms directly by
an inductive family (Dybjer, 1994). Our main motivation for this choice is the beautiful
inductive definition of strongly normalizing terms to follow in Section 5. Since it relies
on a classification of terms into the three shapes introduction, elimination, and weak
head redex, it does not capture all strongly normalizing raw terms, in particular “junk”
terms such as fst (λxx). Of course, statically well-typed terms come also at a cost: for
almost all our predicates on terms we need to show that they are natural in the typing
context, i. e., closed under well-typed renamings. This expense might be compensated
by the extra assistance Agda can give us in proof construction, which is due to the strong
constraints on possible solutions imposed by the rich typing.

Our encoding of well-typed terms follows closely Altenkirch and Reus (1999);
McBride (2006); Benton et al. (2012). We represent typed variables x : Var Γ a by de
Brujin indices, i. e., positions in a typing context Γ : Cxt, which is just a list of types.

Cxt = List Ty

data Var : (Γ : Cxt) (a : Ty) → Set where
zero : ∀{Γ a} → Var (a :: Γ) a
suc : ∀{Γ a b} (x : Var Γ a) → Var (b :: Γ) a

Arguments enclosed in braces, such as Γ, a, and b in the types of the constructors zero
and suc, are hidden and can in most cases be inferred by Agda. If needed, they can
be passed in braces, either as positional arguments (e. g., {∆}) or as named arguments
(e. g., {Γ = ∆}). If ∀ prefixes bindings in a function type, the types of the bound vari-
ables may be omitted. Thus, ∀{Γ a} → A is short for {Γ : Cxt}{a : Ty} → A.

Terms t : Tm Γ a are indexed by a typing context Γ and their type a, guaranteeing
well-typedness and well-scopedness. In the following data type definition, Tm (Γ :Cxt)
shall mean that all constructors uniformly take Γ as their first (hidden) argument.

data Tm (Γ : Cxt) : (a : Ty) → Set where
var : ∀{a} (x : Var Γ a) → Tm Γ a
abs : ∀{a b} (t : Tm (a :: Γ) b) → Tm Γ (a →̂ b)
app : ∀{a b} (t : Tm Γ (a →̂ b)) (u : Tm Γ a) → Tm Γ b
pair : ∀{a b} (t : Tm Γ a) (u : Tm Γ b) → Tm Γ (a ×̂ b)
fst : ∀{a b} (t : Tm Γ (a ×̂ b)) → Tm Γ a
snd : ∀{a b} (t : Tm Γ (a ×̂ b)) → Tm Γ b
next : ∀{a∞} (t : Tm Γ (force a∞)) → Tm Γ (Î a∞)
_∗_ : ∀{a∞ b∞} (t : Tm Γ (Î(a∞ ⇒ b∞))) (u : Tm Γ (Î a∞)) → Tm Γ (Î b∞)

The most natural typing for next and ∗ would be using the defined I_ : Ty → Ty:

next : ∀{a} (t : Tm Γ a) → Tm Γ (I a)
_∗_ : ∀{a b} (t : Tm Γ (I(a →̂ b))) (u : Tm Γ (I a)) → Tm Γ (I b)

However, this would lead to indices like Î delay a and unification problems Agda can-
not solve, since matching on a coinductive constructor like delay is forbidden—it can
lead to a loss of subject reduction (McBride, 2009). The chosen alternative typing,
which parametrizes over a∞ b∞ : ∞Ty rather than a b : Ty, works better in practice.

3.3 Type Equality

Although our coinductive representation of λI types saves us from type variables, type
substitution, and fixed-point unrolling, the question of type equality is not completely
settled. The propositional equality ≡ of Martin-Löf Type Theory is intensional in the
sense that only objects with the same code (modulo definitional equality) are considered
equal. Thus, ≡ is adequate only for finite objects (such as natural numbers and lists) but
not for infinite objects like functions, streams, or λI types.

However, we can define extensional equality or bisimulation on Ty as a mixed
coinductive-inductive relation ≅/∞≅ that follows the structure of Ty/∞Ty (hence, we
reuse the constructor names ×̂, →̂, and Î ).

mutual
data _≅_ : (a b : Ty) → Set where
_×̂_ : ∀{a a’ b b’} (a≅ : a ≅ a’) (b≅ : b ≅ b’) → (a ×̂ b) ≅ (a’ ×̂ b’)
_→̂_ : ∀{a a’ b b’} (a≅ : a’ ≅ a) (b≅ : b ≅ b’) → (a →̂ b) ≅ (a’ →̂ b’)
Î_ : ∀{a∞ b∞} (a≅ : a∞ ∞≅ b∞) → Î a∞ ≅ Î b∞

record _∞≅_ (a∞ b∞ : ∞Ty) : Set where
coinductive
constructor ≅delay
�eld ≅force : force a∞ ≅ force b∞

Ty-equality is indeed an equivalence relation (we omit the standard proof).

≅re� : ∀{a} → a ≅ a
≅sym : ∀{a b} → a ≅ b → b ≅ a
≅trans : ∀{a b c} → a ≅ b → b ≅ c → a ≅ c

However, unlike for ≡ we do not get a generic substitution principle for ≅, but have
to prove it for any function and predicate on Ty. In particular, we have to show that
we can cast a term in Tm Γ a to Tm Γ b if a ≅ b, which would require us to build type
equality at least into Var Γ a. In essence, this would amount to work with setoids across
all our development, which would add complexity without strengthening our result.
Hence, we fall for the shortcut:

It is consistent to postulate that bisimulation implies equality, similarly to the func-
tional extensionality principle for function types. This lets us define the function cast
to convert terms between bisimilar types.

postulate ≅-to-≡ : ∀ {a b} → a ≅ b → a ≡ b

cast : ∀{Γ a b} (eq : a ≅ b) (t : Tm Γ a) → Tm Γ b

We shall require cast in uses of functorial application, to convert a type c∞ : ∞Ty into
something that can be forced into a function type.

Iapp : ∀{Γ c∞ b∞ a} (eq : c∞ ∞≅ (delay a ⇒ b∞))
(t : Tm Γ (Î c∞)) (u : Tm Γ (I a)) → Tm Γ (Î b∞)

Iapp eq t u = cast (Î eq) t ∗ u

3.4 Examples

Following Nakano (2000), we can adapt the Y combinator from the untyped lambda
calculus to define a guarded fixed point combinator:

fix = λ f . (λx. f (x∗next x)) (next (λx. f (x∗next x))).

We construct an auxiliary type Fix a that allows safe self application, since the argument
will only be available "later". This fits with the type we want for the �x combinator,
which makes the recursive instance y in fix (λy. t) available only at the next time slot.

�x : ∀{Γ a} → Tm Γ ((I a →̂ a) →̂ a)

Fix_ : Ty → ∞Ty
force (Fix a) = Î Fix a →̂ a

selfApp : ∀{Γ a} → Tm Γ (Î Fix a) → Tm Γ (I a)
selfApp x = Iapp (≅delay ≅re�) x (next x)

�x = abs (app L (next L))
where
f = var (suc zero)
x = var zero
L = abs (app f (selfApp x))

Another standard example is the type of streams, which we can also define through
corecursion.

mutual
Stream : Ty → Ty
Stream a = a ×̂ Î Stream∞ a

Stream∞ : Ty → ∞Ty
force (Stream∞ a) = Stream a

cons : ∀{Γ a} → Tm Γ a → Tm Γ (I Stream a) → Tm Γ (Stream a)
cons a s = pair a (cast (Î (≅delay ≅re�)) s)

head : ∀{Γ a} → Tm Γ (Stream a) → Tm Γ a
head s = fst s

tail : ∀{Γ a} → Tm Γ (Stream a) → Tm Γ (I Stream a)
tail s = cast (Î (≅delay ≅re�)) (snd s)

Note that tail returns a stream inside the later modality. This ensures that functions
that transform streams have to be causal, i. e., can only have access to the first n elements
of the input when producing the nth element of the output. A simple example is mapping
a function over a stream.

mapS : ∀{Γ a b} → Tm Γ ((a →̂ b) →̂ (Stream a →̂ Stream b))

Which is also better read with named variables.

mapS= λ f . �x (λmapS. λ s. ( f s, mapS∗ tail s))

4 Reduction

In this section, we describe the implementation of parametrized reduction−→n in Agda.
As a prerequisite, we need to define substitution, which in turn depends on renaming
(Benton et al., 2012).

A renaming from context Γ to context ∆ , written ∆ ≤ Γ, is a mapping from variables
of Γ to those of ∆ of the same type a. The function rename lifts such a mapping to terms.

_≤_ : (∆ Γ : Cxt) → Set
_≤_ ∆ Γ = ∀ {a} → Var Γ a → Var ∆ a

rename : ∀ {Γ ∆ : Cxt} {a : Ty} (η : ∆ ≤ Γ) (x : Tm Γ a) → Tm ∆ a

Building on renaming, we define well-typed parallel substitution. From this, we get
the special case of substituting de Bruijn index 0.

subst0 : ∀ {Γ a b} → Tm Γ a → Tm (a :: Γ) b → Tm Γ b

Reduction t −→n t ′ is formalized as the inductive family t 〈n〉⇒β t’ with four con-
structors β... representing the contraction rules and one congruence rule cong to reduce
in subterms.

data _〈_〉⇒β_ {Γ} : ∀ {a} → Tm Γ a → N → Tm Γ a → Set where

β : ∀ {n a b}{t : Tm (a :: Γ) b}{u}
→ app (abs t) u 〈 n 〉⇒β subst0 u t

βfst : ∀ {n a b}{t : Tm Γ a}{u : Tm Γ b}
→ fst (pair t u) 〈 n 〉⇒β t

βsnd : ∀ {n a b}{t : Tm Γ a}{u : Tm Γ b}
→ snd (pair t u) 〈 n 〉⇒β u

βI : ∀ {n a∞ b∞}{t : Tm Γ (force a∞ →̂ force b∞)}{u : Tm Γ (force a∞)}
→ (next t ∗ next {a∞ = a∞} u) 〈 n 〉⇒β (next {a∞ = b∞} (app t u))

cong : ∀ {n n’ ∆ a b t t’ Ct Ct’}{C : NβCxt ∆ Γ a b n n’}
→ (Ct : Ct ≡ C [ t ])
→ (Ct’ : Ct’ ≡ C [ t’ ])
→ (t⇒β : t 〈 n 〉⇒β t’)
→ Ct 〈 n’ 〉⇒β Ct’

The congruence rule makes use of shallow one hole contexts C, which are given by
the following grammar

C ::= λx_ | _u | t _ | (t,_) | (_,u) | fst _ | snd _ | next_ | _∗u | t∗_.

cong says that we can reduce a term, suggestively called Ct, to a term Ct’, if (1) Ct
decomposes into C[t], a context C filled by t, and (2) Ct’ into C[t’], and (3) t reduces
to t’. As witnessed by relation Ct≡C [t], context C : NβCxt Γ ∆ a b n n’ produces a
term Ct : Tm Γ b of depth n’ if filled with a term t : Tm ∆ a of depth n. The depth is
unchanged except for the case next, which increases the depth by 1. Thus, t 〈n〉⇒β t’
can contract every subterm that is under at most n many nexts.

data NβCxt : (∆ Γ : Cxt) (a b : Ty) (n n’ : N) → Set where
abs : ∀{Γ n a b} → NβCxt (a :: Γ) Γ b (a →̂ b) n n
appl : ∀{Γ n a b} (u : Tm Γ a) → NβCxt Γ Γ (a →̂ b) b n n
appr : ∀{Γ n a b} (t : Tm Γ (a →̂ b)) → NβCxt Γ Γ a b n n
pairl : ∀{Γ n a b} (u : Tm Γ b) → NβCxt Γ Γ a (a ×̂ b) n n
pairr : ∀{Γ n a b} (t : Tm Γ a) → NβCxt Γ Γ b (a ×̂ b) n n
fst : ∀{Γ n a b} → NβCxt Γ Γ (a ×̂ b) a n n
snd : ∀{Γ n a b} → NβCxt Γ Γ (a ×̂ b) b n n
next : ∀{Γ n a∞} → NβCxt Γ Γ (force a∞) (Î a∞) n (1 + n)
∗l_ : ∀{Γ n a∞ b∞} (u : Tm Γ (Î a∞)) → NβCxt Γ Γ (Î (a∞ ⇒ b∞)) (Î b∞) n n
∗r_ : ∀{Γ n a∞ b∞}

(t : Tm Γ (Î (a∞ ⇒ b∞))) → NβCxt Γ Γ (Î a∞) (Î b∞) n n

data _≡_[_] {n : N} {Γ : Cxt} : {n’ : N} {∆ : Cxt} {b a : Ty} →

Tm Γ b → NβCxt ∆ Γ a b n n’ → Tm ∆ a → Set

5 Strong Normalization

Classically, a term is strongly normalizing (sn) if there’s no infinite reduction sequence
starting from it. Constructively, the tree of all the possible reductions from an sn term
must be well-founded, or, equivalently, an sn term must be in the accessible part of the
reduction relation. In our case, reduction t 〈n〉⇒β t’ is parametrized by a depth n, thus,
we get the following family of sn-predicates.

data sn (n : N) {a Γ} (t : Tm Γ a) : Set where
acc : (∀ {t’} → t 〈 n 〉⇒β t’ → sn n t’) → sn n t

Van Raamsdonk et al. (1999) pioneered a more explicit characterization of strongly
normalizing terms SN, namely the least set closed under introductions, formation of
neutral (=stuck) terms, and weak head expansion. We adapt their technique from lambda-
calculus to λI; herein, it is crucial to work with well-typed terms to avoid junk like
fst(λx.x) which does not exist in pure lambda-calculus. To formulate a deterministic
weak head evaluation, we make use of the evaluation contexts E : ECxt

E ::= _ u | fst _ | snd _ | _∗u | (next t)∗_.

Since weak head reduction does not go into introductions which include λ -abstraction,
it does not go under binders, leaving typing context Γ fixed.

data ECxt (Γ : Cxt) : (a b : Ty) → Set
data _≅_[_] {Γ : Cxt} : {a b : Ty} → Tm Γ b → ECxt Γ a b → Tm Γ a → Set

Et≅E[t] witnesses the splitting of a term Et into evaluation context E and hole
content t. A generalization of _≅_[_] is PCxt P which additionally requires that all
terms contained in the evaluation context (that is one or zero terms) satisfy predicate
P. This allows us the formulation of P-neutrals as terms of the form ~E[x] for some
~E[_] = E1[. . .En[_]] and a variable x where all immediate subterms satisfy P.

data PCxt {Γ} (P : ∀{c} → Tm Γ c → Set) :
∀ {a b} → Tm Γ b → ECxt Γ a b → Tm Γ a → Set where

appl : ∀ {a b t u} (u : P u) → PCxt P (app t u) (appl u) (t : (a →̂ b))
fst : ∀ {a b t} → PCxt P (fst t) fst (t : (a ×̂ b))
snd : ∀ {a b t} → PCxt P (snd t) snd (t : (a ×̂ b))
∗l_ : ∀ {a∞ b∞ t u} (u : P u) → PCxt P (t ∗ (u : Î a∞) : Î b∞) (∗l u) t
∗r_ : ∀ {a∞ b∞ t u} (t : P (next {a∞ = a∞ ⇒ b∞} t))

→ PCxt P ((next t) ∗ (u : Î a∞) : Î b∞) (∗r t) u

data PNe {Γ} (P : ∀{c} → Tm Γ c → Set) {b} : Tm Γ b → Set where
var : ∀ x → PNe P (var x)
elim : ∀ {a} {t : Tm Γ a} {E Et}

→ (n : PNe P t) (Et : PCxt P Et E t) → PNe P Et

Weak head reduction (whr) is a reduction of the form ~E[t]−→ ~E[t ′] where t 7→ t ′. It is
well-known that weak head expansion (whe) does not preserve sn, e.g., (λx.y)Ω is not
sn even though it contracts to y. In this case, Ω is a vanishing term lost by reduction. If
we require that all vanishing terms in a reduction are sn, weak head expansion preserves
sn. In the following, we define P-whr where all vanishing terms must satisfy P.

data _/_⇒_ {Γ} (P : ∀{c} → Tm Γ c → Set) :
∀ {a} → Tm Γ a → Tm Γ a → Set where

β : ∀ {a b}{t : Tm (a :: Γ) b}{u}
→ (u : P u)
→ P / (app (abs t) u) ⇒ subst0 u t

βfst : ∀ {a b}{t : Tm Γ a}{u : Tm Γ b}
→ (u : P u)
→ P / fst (pair t u) ⇒ t

βsnd : ∀ {a b}{t : Tm Γ a}{u : Tm Γ b}
→ (t : P t)
→ P / snd (pair t u) ⇒ u

βI : ∀ {a∞ b∞}{t : Tm Γ (force (a∞ ⇒ b∞))}{u : Tm Γ (force a∞)}
→ P / (next t ∗ next {a∞ = a∞} u) ⇒ (next {a∞ = b∞} (app t u))

cong : ∀ {a b t t’ Et Et’}{E : ECxt Γ a b}
→ (Et : Et ≅ E [ t ])
→ (Et’ : Et’ ≅ E [ t’ ])
→ (t⇒ : P / t ⇒ t’)
→ P / Et ⇒ Et’

The family of predicates SN n is defined inductively by the following rules—we
allow ourselves set-notation at this semi-formal level:

t ∈ SN n
λxt ∈ SN n

t1, t2 ∈ SN n
(t1, t2) ∈ SN n next t ∈ SN 0

t ∈ SN n
next t ∈ SN (1+n)

t ∈ SNe n
t ∈ SN n

t ′ ∈ SN n t 〈n〉⇒ t ′

t ∈ SN n

The last two rules close SN under neutrals SNe, which is an instance of PNe with
P = SN n, and level-n strong head expansion t 〈n〉⇒ t ′, which is an instance of P-whe
with also P = SN n. We represent the inductive SN in Agda as a sized type (Hughes
et al., 1996; Abel and Pientka, 2013) for the purpose of termination checking certain
inductions on SN later. The assignment of sizes follows the principle that recursive
invokations of SN within a constructor of SN {i} must carry a strictly smaller size
j : Size< i. The mutually defined relations SNe n t (instance of PNe) and strong head
reduction (shr) t 〈n〉⇒ t’ just thread the size argument through. Note that there is a
version i size t 〈n〉⇒ t’ of shr that makes the size argument visible, to be supplied in
case exp.

mutual
data SN {i : Size}{Γ} : (n : N) → ∀ {a} → Tm Γ a → Set where

abs : ∀ {j : Size< i} {a b n}{t : Tm (a :: Γ) b}
→ (t : SN {j} n t)
→ SN n (abs t)

pair : ∀ {j1 j2 : Size< i} {a b n t u}
→ (t : SN {j1} n t) (u : SN {j2} n u)
→ SN n {a ×̂ b} (pair t u)

next0 : ∀ {a∞} {t : Tm Γ (force a∞)}

→ SN 0 {Î a∞} (next t)

next : ∀ {j : Size< i} {a∞ n} {t : Tm Γ (force a∞)}
→ (t : SN {j} n t)
→ SN (1 + n) {Î a∞} (next t)

ne : ∀ {j : Size< i} {a n t}
→ (n : SNe {j} n t)
→ SN n {a} t

exp : ∀ {j1 j2 : Size< i} {a n t t′}
→ (t⇒ : j1 size t 〈 n 〉⇒ t′) (t′ : SN {j2} n t′)
→ SN n {a} t

SNe : ∀ {i : Size} {Γ a} (n : N) → Tm Γ a → Set
SNe {i} n = PNe (SN {i} n)

_size_〈_〉⇒_ : ∀ (i : Size) {Γ a} → Tm Γ a → N → Tm Γ a → Set
i size t 〈 n 〉⇒ t′ = SN {i} n / t ⇒ t′

_〈_〉⇒_ : ∀ {i : Size} {Γ a} → Tm Γ a → N → Tm Γ a → Set
_〈_〉⇒_ {i} t n t’ = SN {i} n / t ⇒ t’

The SN-relations are antitone in the level n. This is one dimension of the Kripke
worlds in our model (see next section).

mapSN : ∀ {m n} → m ≤N n → ∀ {Γ a}{t : Tm Γ a} → SN n t → SN m t

mapSNe : ∀ {m n} → m ≤N n → ∀ {Γ a}{t : Tm Γ a} → SNe n t → SNe m t
map⇒ : ∀ {m n} → m ≤N n → ∀ {Γ a}{t t’ : Tm Γ a} → t 〈 n 〉⇒ t’ → t 〈 m 〉⇒ t’

The other dimension of the Kripke worlds is the typing context; our notions are also
closed under renaming (and even undoing of renaming). Besides renameSN, we have
analogous lemmata renameSNe and rename⇒.

renameSN : ∀ {n a ∆ Γ} (ρ : ∆ ≤ Γ) {t : Tm Γ a} →

SN n t → SN n (rename ρ t)

fromRenameSN : ∀{n a Γ ∆} (ρ : ∆ ≤ Γ) {t : Tm Γ a} →

SN n (rename ρ t) → SN n t

A consequence of fromRenameSN is that t ∈ SN n iff t x∈ SN n for some variable x.
(Consider t = λy. t ′ and t x 〈n〉⇒ t ′[y/x].) This property is essential for the construction
of the function space on sn sets (see next section).

absVarSN : ∀{Γ a b n}{t : Tm (a :: Γ) (a →̂ b)} →

app t (var zero) ∈ SN n → t ∈ SN n

6 Soundness

A well-established technique (Tait, 1967) to prove strong normalization is to model each
type a as a set A = JaK of sn terms. Each so-called semantic type A should contain the
variables in order to interpret open terms by themselves (using the identity valuation).
To establish the conditions of semantic types compositionally, the set A needs to be
saturated, i. e., contain SNe (rather than just the variables) and be closed under strong
head expansion (to entertain introductions).

As a preliminary step towards saturated sets we define sets of well-typed terms in
an arbitrary typing context but fixed type, TmSet a. We also define shorthands for the
largest set, set inclusion and closure under expansion.

TmSet : (a : Ty) → Set1
TmSet a = {Γ : Cxt} (t : Tm Γ a) → Set

[>] : ∀{a} → TmSet a
[>] t = >

_⊆_ : ∀{a} (AA′ : TmSet a) → Set
A ⊆ A′ = ∀{Γ}{t : Tm Γ _} → A t → A′ t

Closed : ∀ (n : N) {a} (A : TmSet a) → Set
Closed n A = ∀{Γ}{t t’ : Tm Γ _} → t 〈 n 〉⇒ t’ → A t’ → A t

For each type constructor we define a corresponding operation on TmSets. The
product is simply pointwise through the use of the projections.

_[×]_ : ∀{a b} → TmSet a → TmSet b → TmSet (a ×̂ b)
(A [×] B) t = A (fst t) × B (snd t)

For function types we are forced to use a Kripke-style definition, quantifying over
all possible extended contexts ∆ makes A [→] B closed under renamings.

_[→]_ : ∀{a b} → TmSet a → TmSet b → TmSet (a →̂ b)
(A [→] B) {Γ} t = ∀{∆} (ρ : ∆ ≤ Γ) → ∀ {u} → A u → B (app (rename ρ t) u)

The TmSet for the later modality is indexed by the depth. The first two constructors
are for terms in the canonical form next t, at depth zero we impose no restriction on t,
otherwise we use the given set A. The other two constructors are needed to satisfy the
properties we require of our saturated sets.

data [I] {a∞} (A : TmSet (force a∞)) {Γ} : (n : N) → Tm Γ (Î a∞) → Set where
next0 : ∀ {t : Tm Γ (force a∞)} → [I]A zero (next t)
next : ∀ {n}{t : Tm Γ (force a∞)} (t :A t) → [I]A (suc n) (next t)
ne : ∀ {n}{t : Tm Γ (Î a∞)} (n : SNe n t) → [I]A n t
exp : ∀ {n}{t t’ : Tm Γ (Î a∞)}

(t⇒ : t 〈 n 〉⇒ t’) (t : [I]A n t’) → [I]A n t

The particularity of our saturated sets is that they are indexed by the depth, which
in our case is needed to state the usual properties. In particular if a term belongs to a

saturated set it is also a member of SN, which is what we need for strong normalization.
In addition we require them to be closed under renaming, since we are dealing with
terms in a context.

record IsSAT (n : N) {a} (A : TmSet a) : Set where
�eld
satSNe : SNe n ⊆ A
satSN : A ⊆ SN n
satExp : Closed n A
satRename : ∀ {Γ ∆} (ρ : ∆ ≤ Γ) → ∀ {t} → A t → A (rename ρ t)

record SAT (a : Ty) (n : N) : Set1 where
�eld
satSet : TmSet a
satProp : IsSAT n satSet

For function types we will also need a notion of a sequence of saturated sets up to a
specified maximum depth n.

SAT≤ : (a : Ty) (n : N) → Set1
SAT≤ a n = ∀ {m} → m ≤N n → SAT a m

To help Agda’s type inference, we also define a record type for membership of a
term into a saturated set.

record _∈_ {a n Γ} (t : Tm Γ a) (A : SAT a n) : Set where
constructor �_
�eld �_ : satSet A t

_∈〈_〉_ : ∀ {a n Γ} (t : Tm Γ a) {m} (m≤n : m ≤N n) (A : SAT≤ a n) → Set
t ∈〈 m≤n 〉 A = t ∈ A m≤n

Given the lemmas about SN shown so far we can lift our operations on TmSet to
saturated sets and give the semantic version of our term constructors.

For function types we need another level of Kripke-style generalization to smaller
depths, so that we can maintain antitonicity.

_J→K_ : ∀ {n a b} (A : SAT≤ a n) (B : SAT≤ b n) → SAT (a →̂ b) n
A J→K B = record
{ satSet = λ t → ∀ m (m≤n : m ≤N _) → (A m≤n [→]B m≤n) t
; satProp = record
{ satSN = CSN

; satSNe = CSNe
; satExp = CExp
; satRename = CRename

}
}
where
module A = SAT≤ A
module B = SAT≤ B

A = A .satSet
B = B.satSet

C : TmSet (_ →̂ _)
C t = ∀ m (m≤n : m ≤N _) → (A m≤n [→]B m≤n) t

CSN : C ⊆ SN _
CSN t = fromRenameSN suc (absVarSN
(B.satSN ≤N.re� (t _ ≤N.re� suc (A .satSNe ≤N.re� (var zero)))))

CSNe : SNe _ ⊆ C
CSNe n m m≤n ρ u =

B.satSNe m≤n (sneApp (mapSNe m≤n (renameSNe ρ n)) (A .satSN m≤n u))

CExp : ∀{Γ}{t t’ : Tm Γ _} → t 〈 _ 〉⇒ t’ → C t’ → C t
CExp t⇒ t m m≤n ρ u =

B.satExp m≤n ((cong (appl _) (appl _) (map⇒ m≤n (rename⇒ ρ t⇒)))) (t m m≤n ρ u)

CRename : {Γ ∆ : List Ty} (ρ : ∆ ≤ Γ) {t : Tm Γ _} → C t → C (rename ρ t)
CRename = λ ρ {t} t m m≤n ρ’ {u} u →

≡.subst (λ t1 → B {m} m≤n (app t1 u)) (subst-• ρ’ ρ t) (t m m≤n (ρ’ •s ρ) u)

The proof of inclusion into SN first derives that app (rename suc t) (var zero) is in
SN through the inclusion of neutral terms into A and the inclusion of B into SN, then
proceeds to strip away first (var zero) and then (rename suc), so that we are left with
the original goal SN n t. Renaming t with suc is necessary to be able to introduce the
fresh variable zero of type a.

The types of semantic abstraction and application are somewhat obfuscated because
they need to mention the upper bounds and the renamings.

JabsK : ∀ {n a b} {A : SAT≤ a n} {B : SAT≤ b n} {Γ} {t : Tm (a :: Γ) b} →

(∀ {m} (m≤n : m ≤N n) {∆} (ρ : ∆ ≤ Γ) {u : Tm ∆ a} →

u ∈〈 m≤n 〉 A → (subst0 u (subst (lifts ρ) t)) ∈〈 m≤n 〉B)
→ abs t ∈ (A J→K B)

(� JabsK {A = A }{B = B} t) m m≤n ρ u =
SAT≤.satExp B m≤n (β (SAT≤.satSN A m≤n u)) (� t m≤n ρ (� u))

JappK : ∀ {n a b}{A : SAT≤ a n}{B : SAT≤ b n}{Γ}{t : Tm Γ (a →̂ b)}{u : Tm Γ a}
→ t ∈ (A J→K B) → u ∈〈 ≤N.re� 〉 A → app t u ∈〈 ≤N.re� 〉B

JappK {B = B} {u = u} (� t) (� u) = ≡.subst (λ t → app t u ∈〈 ≤N.re� 〉B) renId
(� t _ ≤N.re� id u)

The TmSet for product types is directly saturated, inclusion into SN uses a lemma
to derive SN n t from SN n (fst t), which follows from A ⊆ SN.

_J×K_ : ∀ {n a b} (A : SAT a n) (B : SAT b n) → SAT (a ×̂ b) n
A J×K B = record
{ satSet = satSet A [×] satSet B
; satProp = record
{ satSNe = CSNe
; satSN = CSN

; satExp = CExp
; satRename = λ ρ x → satRename A ρ (proj1 x) , satRename B ρ (proj2 x)
}

}
where
A = satSet A
B = satSet B
C : TmSet _
C = A [×]B

CSNe : SNe _ ⊆ C
CSNe n = satSNe A (elim n fst)

, satSNe B (elim n snd)

CSN : C ⊆ SN _
CSN (t , u) = bothProjSN (satSN A t) (satSN B u)

CExp : ∀{Γ}{t t’ : Tm Γ _} → t 〈 _ 〉⇒ t’ → C t’ → C t
CExp t⇒ (t , u) = satExp A (cong fst fst t⇒) t

, satExp B (cong snd snd t⇒) u

Semantic introduction JpairK : t1 ∈ A → t2 ∈ B → pair t1 t2 ∈ (A J×K B) and
eliminations JfstK : t ∈ (A J×K B) → fst t ∈ A and JsndK : t ∈ (A J×K B) → snd t ∈

B for pairs are straightforward.

JpairK : ∀ {n a b} {A : SAT a n} {B : SAT b n} {Γ} {t1 : Tm Γ a} {t2 : Tm Γ b}
→ t1 ∈ A → t2 ∈ B → pair t1 t2 ∈ (A J×K B)

� JpairK {A = A } {B = B} (� t) (� u) = satExp A (βfst (satSN B u)) t
, satExp B (βsnd (satSN A t)) u

JfstK : ∀ {n a b} {A : SAT a n} {B : SAT b n} {Γ} {t : Tm Γ (a ×̂ b)}
→ t ∈ (A J×K B) → fst t ∈ A

JfstK t = � (proj1 (� t))

JsndK : ∀ {n a b} {A : SAT a n} {B : SAT b n} {Γ} {t : Tm Γ (a ×̂ b)}
→ t ∈ (A J×K B) → snd t ∈ B

JsndK t = � (proj2 (� t))

The later modality is going to use the saturated set for its type argument at the
preceeding depth, we encode this fact through the type SATpred.

SATpred : (a : Ty) (n : N) → Set1
SATpred a zero = >
SATpred a (suc n) = SAT a n

SATpredSet : {n : N}{a : Ty} → SATpred a n → TmSet a
SATpredSet {zero} A = [>]
SATpredSet {suc n} A = satSet A

Since the cases for [I]_ are essentially a subset of those for SN, the proof of inclu-
sion into SN goes by induction and the inclusion of A into SN.

JIK_ : ∀{n a∞} (A : SATpred (force a∞) n) → SAT (Î a∞) n
JIK_ {n} {a∞} A = record
{ satSet = [I] (SATpredSet A ) n

; satProp = record
{ satSNe = ne
; satSN = CSN A
; satExp = exp
; satRename = CRen A
}

}
where
C : ∀ {n} (A : SATpred (force a∞) n) → TmSet (Î a∞)
C {n} A = [I] (SATpredSet A ) n

CSN : ∀ {n} (A : SATpred (force a∞) n) → C {n} A ⊆ SN n
CSN A next0 = next0
CSN A (next t) = next (satSN A t)
CSN A (ne n) = ne n
CSN A (exp t⇒ t) = exp t⇒ (CSN A t)

CRen : ∀ {n} (A : SATpred (force a∞) n) → ∀ {Γ ∆} (ρ : Γ ≤ ∆) →

∀ {t} → C {n} A t → C {n} A (subst ρ t)
CRen A ρ next0 = next0
CRen A ρ (next t) = next (satRename A ρ t)
CRen A ρ (ne n) = ne (renameSNe ρ n)
CRen A ρ (exp t⇒ t) = exp (rename⇒ ρ t⇒) (CRen A ρ t)

Following Section 3 we can assemble the combinators for saturated sets into a se-
mantics for the types of λI. The definition of J_K_ proceeds by recursion on the induc-
tive part of the type, and otherwise by well-founded recursion on the depth. Crucially
the interpretation of the later modality only needs the interpretation of its type parame-
ter at a smaller depth, which is then decreasing exactly when the representation of types
becomes coinductive and would no longer support recursion.

J_K≤ : (a : Ty) {n : N} → ∀ {m} → m ≤N n → SAT a m

J_K_ : (a : Ty) (n : N) → SAT a n
J a →̂ b K n = J a K≤ {n} J→K J b K≤ {n}
J a ×̂ b K n = J a K n J×K J b K n
J Î a∞ K n = JIK P n
where
P : ∀ n → SATpred (force a∞) n
P zero = _
P (suc n) = J force a∞ K n

Well-founded recursion on the depth is accomplished through the auxiliary defi-
nition J_K≤ which recurses on the inequality proof. It is however straightforward to
convert in and out of the original interpretation, or between different upper bounds.

in≤ : ∀ a {n m} (m≤n : m ≤N n) → satSet (J a K m) ⊆ satSet (J a K≤ m≤n)
out≤ : ∀ a {n m} (m≤n : m ≤N n) → satSet (J a K≤ m≤n) ⊆ satSet (J a K m)

coerce≤ : ∀ a {n n’ m} (m≤n : m ≤N n) (m≤n’ : m ≤N n’)
→ satSet (J a K≤ m≤n) ⊆ satSet (J a K≤ m≤n’)

As will be necessary later for the interpretation of next, the interpretation of types
is also antitone. For most types this follows by recursion, while for function types anti-
tonicity is embedded in their semantics and we only need to convert between different
upper bounds.

mapJ_K : ∀ a {m n} → m ≤N n → satSet (J a K n) ⊆ satSet (J a K m)

mapJ a →̂ b K m≤n t = λ l l≤m ρ u → let l≤n = ≤N.trans l≤m m≤n in
coerce≤ b l≤n l≤m (t l l≤n ρ (coerce≤ a l≤m l≤n u))

mapJ a ×̂ b K m≤n (t , u) = mapJ a K m≤n t , mapJ b K m≤n u
mapJ Î a∞ K m≤n (ne n) = ne (mapSNe m≤n n)
mapJ Î a∞ K m≤n (exp t⇒ t) = exp (map⇒ m≤n t⇒) (mapJ Î a∞ K m≤n t)
mapJ Î a∞ K {m = zero} m≤n next0 = next0
mapJ Î a∞ K {m = suc m} () next0
mapJ Î a∞ K {m = zero} m≤n (next _) = next0
mapJ Î a∞ K {m = suc m} m≤n (next t) = next (mapJ force a∞ K (pred≤N m≤n) t)

Typing contexts are interpreted as predicates on substitutions. These predicates in-
herit antitonicity and closure under renaming. Semantically sound substitutions act as
environments θ. We will need Ext to extend the environment for the interpretation of
lambda abstractions.

J_KC : ∀ Γ {n} → ∀ {∆} (σ : Subst Γ ∆) → Set
J Γ KC {n} σ = ∀ {a} (x : Var Γ a) → σ x ∈ J a K n

Map : ∀ {m n} → (m≤n : m ≤N n) →

∀ {Γ ∆} {σ : Subst Γ ∆} (θ : J Γ KC {n} σ) → J Γ KC {m} σ

Map m≤n θ {a} x = mapJ a K∈ m≤n (θ x)

Rename : ∀ {n ∆ ∆ ’} → (ρ : Ren ∆ ∆ ’) →

∀ {Γ}{σ : Subst Γ ∆} (θ : J Γ KC {n} σ) →

J Γ KC (ρ •s σ)
Rename ρ θ {a} x = � satRename (J a K _) ρ (� θ x)

Ext : ∀ {a n ∆ Γ} {t : Tm ∆ a} → (t : t ∈ J a K n) →

∀ {σ : Subst Γ ∆} (θ : J Γ KC σ) → J a :: Γ KC (t ::s σ)
Ext t θ (zero) = t
Ext t θ (suc x) = θ x

The soundness proof, showing that every term of λI is a member of our saturated
sets and so a member of SN, is now a simple matter of interpreting each operation in
the language to its equivalent in the semantics that we have defined so far.

sound : ∀ {n a Γ} (t : Tm Γ a) {∆} {σ : Subst Γ ∆} →

(θ : J Γ KC {n} σ) → subst σ t ∈ J a K n
sound (var x) θ = θ x
sound (abs t) θ = JabsK {t = t} λ m≤n ρ u →

� in≤ _ m≤n (� sound t (Ext (� out≤ _ m≤n (� u)) (Rename ρ (Map m≤n θ))))
sound (app t u) θ = JappK (sound t θ) (sound u θ)
sound (pair t u) θ = JpairK (sound t θ) (sound u θ)
sound (fst t) θ = JfstK (sound t θ)
sound (snd t) θ = JsndK (sound t θ)
sound (t ∗ u) θ = J∗K (sound t θ) (sound u θ)
sound {zero} (next t) θ = � next0
sound {suc n} (next t) θ = � (next (� sound t (Map n≤sn θ)))

The interpretation of next depends on the depth, at zero we are done, at suc n we
recurse on the subterm at depth n, using antitonicity to Map the current environment to
depth n as well. In fact without next we would not have needed antitonicity at all since
there would have been no way to embed a term from a smaller depth into a larger one.

7 SN correctness

To complete our strong normalization proof we need to show that SN is included in the
characterization of strong normalization as a well-founded predicate sn.

fromSN : ∀ {i} {Γ} {n : N} {a} {t : Tm Γ a} →

SN {i} n t → sn n t

The cases for canonical and neutral forms are straightforward, since no reduction
can happen at the top of the expression and we cover the others through the induction
hypotheses.

fromSNe : ∀ {i Γ n a} {t : Tm Γ a} →

SNe {i} n t → sn n t

fromSN (ne n) = fromSNe n
fromSN (abs t) = abssn (fromSN t)
fromSN (pair t u) = pairsn (fromSN t) (fromSN u)
fromSN next0 = next0sn
fromSN (next t) = nextsn (fromSN t)
fromSN (exp t⇒ t1) = acc (expsn t⇒ t1 (fromSN t1))

The expansion case is more challenging instead, we can not in fact prove expsn by
induction directly.

expsn : ∀ {i j Γ n a} {t th to : Tm Γ a} →

i size t 〈 n 〉⇒ th → SN {j} n th → sn n th →

t 〈 n 〉⇒β to → sn n to

We can see the problem by looking at one of the congruence cases, in particular
reduction on the left of an application. There we would have t u ∈ sn, tht1 and tβ t2, and
need to prove t2 u ∈ sn. By induction we could obtain t2 ∈ sn but then there would be no
easy way to obtain t2 u ∈ sn, since strong normalization is not closed under application.

The solution is to instead generalize the statement to work under a sequence of head
reduction evaluation contexts. We represent such sequences with the type ECxt*, and
denote their application to a term with the operator _[_]*.

expsnCxt : ∀ {i j Γ n a b} {t th to : Tm Γ a} →

(Es : ECxt* Γ a b) → i size t 〈 n 〉⇒ th →

SN {j} n (Es [ th ]*) → sn n (Es [ th ]*) →

t 〈 n 〉⇒β to → sn n (Es [ to ]*)
expsn t⇒ t t t⇒β = expsnCxt [] t⇒ t t t⇒β

In this way the congruence cases are solved just by induction with a larger context.

expsnCxt E (cong (appl u) (appl .u) th⇒) th th (cong (appl .u) (appl .u) t⇒)
= expsnCxt (appl u :: E) th⇒ th th t⇒

This generalization however affects the lemmata that handle the reduction cases,
which also need to work under a sequence of evaluation contexts. Fortunately the addi-
tion of a premise E[z] ∈ sn, about an unrelated term z, allows to conveniently handle all
the reductions that target the context.

βIsn : ∀ {n Γ b} {a∞ b∞} {z} {t : Tm Γ (force (a∞ ⇒ b∞))} {u : Tm Γ (force a∞)}
(E : ECxt* Γ (Î b∞) b) → sn (suc n) (E [ z ]*) →

sn n t → sn n u → sn (suc n) (E [ next (app t u) ]*) →

sn (suc n) (E [ next t ∗ next {a∞ = a∞} u ]*)

βfstsn : ∀ {n Γ b} {a c} {z} {t : Tm Γ b} {u : Tm Γ a}
(E : ECxt* Γ b c) → sn n (E [ z ]*) →

sn n t → sn n u → sn n (E [ t ]*) →

sn n (E [ fst (pair t u) ]*)

βsndsn : ∀ {n Γ b} {a c} {z} {t : Tm Γ b} {u : Tm Γ a}
(E : ECxt* Γ b c) → sn n (E [ z ]*) →

sn n t → sn n u → sn n (E [ t ]*) →

sn n (E [ snd (pair u t) ]*)

βsn : ∀ {i n a b c Γ} {u : Tm Γ a} {t : Tm (a :: Γ) b}{z}
(Es : ECxt* Γ b c) → sn n (Es [ z ]*) →

sn n t → SN {i} n (Es [ subst0 u t ]*) → sn n u →

sn n (Es [ app (abs t) u ]*)

8 Conclusions

In this paper, we presented a family of strongly-normalizing reduction relations for
simply-typed lambda calculus with Nakano’s modality for recursion. Using a similar
stratification, Krishnaswami and Benton (2011a) have shown weak normalization using
hereditary substitutions, albeit for a system without recursive types.

Our Agda formalization uses a saturated sets semantics based on an inductive no-
tion of strong normalization. Herein, we represented recursive types as infinite type
expressions and terms as intrinsically well-typed ones.

Our treatment of infinite type expressions was greatly simplified by adding an exten-
sionality axiom for the underlying coinductive type to Agda’s type theory. This would
not have been necessary in a more extensional theory such as Observational Type The-
ory (Altenkirch et al., 2007) as shown in (McBride, 2009). Possibly Homotopy Type
Theory (UnivalentFoundations, 2013) would also address this problem, but there the
status of coinductive types is yet unclear.

For the future, we would like to investigate how to incorporate guarded recursive
types into a dependently-typed language, and how they relate to other approaches like
coinduction with sized types, for instance.

